Soluzioni del compito di Metodi Matematici e Calcolo per Ingegneria del 1 febbraio 2005

Enrico Bertolazzi

Trasformata di Laplace

```
> restart:
  with(inttrans) :
```

Data la seguente funzione

```
> f := t -> 1+t+t^2 ;
```

$$f := t \to 1 + t + t^2$$

Usando le regole di trasformazione calcolare le trasformate delle funzioni

```
> f1 := t -> f(t)*exp(-3*t) ;
  f2 := t -> f(6*t) ;
  f3 := t -> D(f)(t) ;
```

$$f1 := t \to f(t)\, \mathbf{e}^{(-3\, t)}$$

$$f2 := t \to f(6\, t)$$

$$f3 := t \to \mathrm{D}(f)(t)$$

Trasformate con le primitive Maple

```
> laplace(f(t),t,s);
  laplace(f1(t),t,s);
  laplace(f2(t),t,s);
  laplace(f3(t),t,s);
```

$$\frac{s^2 + s + 2}{s^3}$$

$$\frac{s^2 + 7\, s + 14}{(s + 3)^3}$$

$$\frac{s^2 + 6\, s + 72}{s^3}$$

$$\frac{2 + s}{s^2}$$

Soluzione di ODE con Laplace

```
> restart:
  with(inttrans) :
```

Data la seguente equazione differenziale

```
> ode := diff(y(x),x,x)+diff(y(x),x)-y(x)=exp(-x) ;
```

$$ode := \left(\frac{d^2}{dx^2}\, y(x)\right) + \left(\frac{d}{dx}\, y(x)\right) - y(x) = \mathbf{e}^{(-x)}$$

Con dato iniziale

```
> y0, yp0 := 1,2 ;
```

$$y0, yp0 := 1, 2$$

Calcolare le soluzioni con le trasformate di Laplace.

Trasformo la equazione differenziale con la trasformata di Laplace

```
> sode := laplace(ode,x,s) ;
```

$$sode := s^2\, laplace(y(x), x, s) - \mathrm{D}(y)(0) - s\, y(0) + s\, laplace(y(x), x, s) - y(0) - laplace(y(x), x, s) = \frac{1}{1+s}$$

Risolvo la equazione per y(s)

```
> lode := isolate(sode,laplace(y(x),x,s));
```

$$lode := laplace(y(x), x, s) = \frac{\frac{1}{1+s} + \mathrm{D}(y)(0) + s\, y(0) + y(0)}{s^2 + s - 1}$$

Applico le condizioni iniziali ottenendo y(s)

```
> ly := subs( y(0)=y0,D(y)(0)=yp0,rhs(lode)) ;
```

$$ly := \frac{\frac{1}{1+s} + 3 + s}{s^2 + s - 1}$$

Espansione in fratti semplici

```
> convert(ly, parfrac, s);
```

$$\frac{2\,s+3}{s^2+s-1} - \frac{1}{1+s}$$

Antitrasformo per ottenere la equazione y(x)

```
> res := invlaplace(ly,s,t) ;
```

$$res := -\mathbf{e}^{(-t)} + \frac{2}{5}\,\mathbf{e}^{\left(-\frac{1}{2}t\right)} \left(5\cosh\left(\frac{1}{2}t\sqrt{5}\right) + 2\sqrt{5}\,\sinh\left(\frac{1}{2}t\sqrt{5}\right)\right)$$

Scritto in forma di esponenziali

```
> collect(simplify(expand(convert(res,exp)),power),exp) ;
```

$$\left(1 - \frac{2}{5}\sqrt{5}\right)\mathbf{e}^{\left(-\frac{1}{2}t - \frac{1}{2}t\sqrt{5}\right)} - \mathbf{e}^{(-t)} + \left(1 + \frac{2}{5}\sqrt{5}\right)\mathbf{e}^{\left(-\frac{1}{2}t + \frac{1}{2}t\sqrt{5}\right)}$$

Soluzione di un sistema di ODE con Laplace

```
> restart:
  with(inttrans) :
```

Dato il seguente sistema di equazioni differenziali

```
> ode1 := diff(y(x),x)+diff(z(x),x)-y(x)=1 ;
  ode2 := diff(y(x),x)-diff(z(x),x)-z(x)=2 ;
```

$$ode1 := \left(\frac{d}{dx}\,y(x)\right) + \left(\frac{d}{dx}\,z(x)\right) - y(x) = 1$$

$$ode2 := \left(\frac{d}{dx}\,y(x)\right) - \left(\frac{d}{dx}\,z(x)\right) - z(x) = 2$$

Con dato iniziale

```
> y0, z0 := 1,2 ;
```

$$y0, z0 := 1, 2$$

Calcolare le soluzioni con le trasformate di Laplace.

Trasformo le equazione differenziale con la trasformata di Laplace

```
> sode1 := laplace(ode1,x,s) ;
  sode2 := laplace(ode2,x,s) ;
```

$$sode1 := s\,laplace(y(x), x, s) - y(0) + s\,laplace(z(x), x, s) - z(0) - laplace(y(x), x, s) = \frac{1}{s}$$

$$sode2 := s\,laplace(y(x), x, s) - y(0) - s\,laplace(z(x), x, s) + z(0) - laplace(z(x), x, s) = \frac{2}{s}$$

Risolvo la equazione per y(s), z(s)

```
> RES := solve({sode1,sode2},{laplace(y(x),x,s),laplace(z(x),x,s)});
```

$$RES := \left\{ laplace(y(x), x, s) = \frac{2\,y(0)\,s^2 + 3\,s + y(0)\,s + z(0)\,s + 1}{(2\,s^2 - 1)\,s}, \right.$$
$$\left. laplace(z(x), x, s) = \frac{-s + y(0)\,s - z(0)\,s + 2 + 2\,z(0)\,s^2}{s\,(2\,s^2 - 1)} \right\}$$

Applico le condizioni iniziali ottenendo y(s), z(s)

```
> SOL :=
  subs(RES,y(0)=y0,z(0)=z0,<laplace(y(x),x,s),laplace(z(x),x,s)>);
```

$$SOL := \begin{bmatrix} \frac{2\,s^2 + 6\,s + 1}{(2\,s^2 - 1)\,s} \\ \\ \frac{-2\,s + 2 + 4\,s^2}{s\,(2\,s^2 - 1)} \end{bmatrix}$$

Antitrasformo per ottenere y(x), z(x)

```
> yy := invlaplace(SOL[1],s,x) ;
  zz := invlaplace(SOL[2],s,x) ;
```

$$yy := -1 + 2\cosh\left(\frac{1}{2}\sqrt{2}\,x\right) + 3\sqrt{2}\sinh\left(\frac{1}{2}\sqrt{2}\,x\right)$$

$$zz := -2 + 4\cosh\left(\frac{1}{2}\sqrt{2}\,x\right) - \sqrt{2}\sinh\left(\frac{1}{2}\sqrt{2}\,x\right)$$

Scritto con soli esponenziali

```
> collect(simplify(expand(convert(yy,exp)),power),exp) ;
  collect(simplify(expand(convert(zz,exp)),power),exp) ;
```

$$\left(1 + \frac{3}{2}\sqrt{2}\right) e^{\left(\frac{1}{2}\sqrt{2}\,x\right)} - 1 + \left(1 - \frac{3}{2}\sqrt{2}\right) e^{\left(-\frac{1}{2}\sqrt{2}\,x\right)}$$

$$\left(2 - \frac{1}{2}\sqrt{2}\right) e^{\left(\frac{1}{2}\sqrt{2}\,x\right)} - 2 + \left(2 + \frac{1}{2}\sqrt{2}\right) e^{\left(-\frac{1}{2}\sqrt{2}\,x\right)}$$

Espansione in fratti semplici per controllo

```
> convert(SOL[1], parfrac, s);
  convert(SOL[2], parfrac, s);
```

$$-\frac{1}{s} + \frac{4s + 6}{2s^2 - 1}$$

$$-\frac{2}{s} + \frac{-2 + 8s}{2s^2 - 1}$$

Soluzione di ricorrenza con trasformata zeta

```
> restart:
```

Risolvere la seguente ricorrenza

```
> RIC := f(n+2) = f(n+1) + 2*f(n)+1;
```

$$RIC := f(n+2) = f(n+1) + 2\,f(n) + 1$$

Con dato iniziale

```
> INI := f(0)=1,f(1)=1 ;
```

$$INI := f(0) = 1, f(1) = 1$$

Usando le primitive si maple:

```
> rsolve({RIC,INI}, f(k));
```

$$\frac{1}{2}(-1)^k + 2^k - \frac{1}{2}$$

Usando la Z-trasformata (occhio negli appunti abbiamo Z(f) = sum(f(n)*z^n mentre maple usa Z(f) = sum(f(n)*w^{-n}

Quindi per confrontare le trasformate bisogna trasfromare w -> 1/z;

```
> zRIC := ztrans(RIC,n,w):
  simplify(subs( ztrans(f(n),n,w)=f(z), w=1/z, zRIC)) ;
```

$$-\frac{-f(z)+f(0)+f(1)\,z}{z^2}=\frac{-f(z)-f(z)\,z+f(0)-f(0)\,z+2\,f(z)\,z^2-z}{z\,(-1+z)}$$

Ricavo f(w) [f(1/z)]

```
> zRICrhs := isolate(zRIC,ztrans(f(n),n,w)):
  simplify(subs( ztrans(f(n),n,w)=f(z), w=1/z, zRICrhs)) ;
```

$$f(z)=\frac{f(0)-2\,f(0)\,z+f(1)\,z-f(1)\,z^2+f(0)\,z^2+z^2}{(-1+z)\,(-1+z+2\,z^2)}$$

Applico le condizioni iniziali

```
> zRICrhsINI := subs(INI,zRICrhs):
  simplify(subs( ztrans(f(n),n,w)=f(z), w=1/z, zRICrhsINI)) ;
```

$$f(z)=\frac{1-z+z^2}{(-1+z)\,(-1+z+2\,z^2)}$$

Conversione in fratti semplici

```
> convert(%, parfrac);
```

$$f(z)=-\frac{1}{2\,z-1}+\frac{1}{2\,(z+1)}+\frac{1}{2\,(-1+z)}$$

Inversione della Z-trasformata

```
> invztrans(zRICrhsINI,w,k) ;
```

$$f(k)=\frac{1}{2}\,(-1)^k-\frac{1}{2}+2^k$$

Soluzione di sistema non lineare con Newton

```
> restart:
  with(VectorCalculus):
Warning, the assigned names `<,>` and `<|>` now have a global binding

Warning, these protected names have been redefined and unprotected:
`*`, `+`, `.`, D, Vector, diff, int, limit, series
```

Sistema non lineare

```
> f := x-y-1 ;
  g := x^2-y^2-4 ;
```

$$f:=x-y-1$$

$$g:=x^2-y^2-4$$

Soluzione esatta

```
> solve({f,g},{x,y}) ;
```

$$\left\{ x = \frac{5}{2}, y = \frac{3}{2} \right\}$$

Matrice Jacobiano

```
> J := Jacobian([f,g],[x,y]) ;
```

$$J := \begin{bmatrix} 1 & -1 \\ 2\,x & -2\,y \end{bmatrix}$$

Schema di Newton

```
> Newton_update := <x,y>-J^(-1).<f,g> ;
```

$$Newton\_update := \left(x - \frac{y\,(x-y-1)}{y-x} + \frac{x^2-y^2-4}{2\,(y-x)} \right) e_x + \left(y - \frac{x\,(x-y-1)}{y-x} + \frac{x^2-y^2-4}{2\,(y-x)} \right) e_y$$

Schema di Newton per questo sistema non lineare

```
> x[k+1]=simplify(subs(x=x[k],y=y[k],Newton_update[1])) ;
  y[k+1]=simplify(subs(x=x[k],y=y[k],Newton_update[2])) ;
```

$$x_{k+1} = \frac{x_k^2 - y_k^2 - 2\,y_k + 4}{2\,(-y_k + x_k)}$$

$$y_{k+1} = \frac{-y_k^2 + x_k^2 - 2\,x_k + 4}{2\,(-y_k + x_k)}$$

Tre iterate a partire da (-1,1)

```
> x[0],y[0]:= -1,1 ;
```

$$x_0, y_0 := -1, 1$$

Prima iterata

```
> x[1] := subs(x=x[0],y=y[0],Newton_update[1]) ;
  y[1] := subs(x=x[0],y=y[0],Newton_update[2]) ;
```

$$x_1 := \frac{-1}{2}$$

$$y_1 := \frac{-3}{2}$$

Seconda iterata

```
> x[2] := subs(x=x[1],y=y[1],Newton_update[1]) ;
  y[2] := subs(x=x[1],y=y[1],Newton_update[2]) ;
```

$$x_2 := \frac{5}{2}$$

$$y_2 := \frac{3}{2}$$

Terza iterata

```
> x[3] := subs(x=x[2],y=y[2],Newton_update[1]) ;
  y[3] := subs(x=x[2],y=y[2],Newton_update[2]) ;
```

$$x_3 := \frac{5}{2}$$

$$y_3 := \frac{3}{2}$$

Problema di Minimo Vincolato

```
> restart:
  with(LinearAlgebra):
  with(Optimization):
  with(VectorCalculus):
Warning, the names `&x`, CrossProduct and DotProduct have been rebound

Warning, the assigned names `<,>` and `<|>` now have a global binding

Warning, these protected names have been redefined and unprotected:
`*`, `+`, `.`, D, Vector, diff, int, limit, series
```

Minimizzare la seguente funzione

```
> f := x^2+y^2+z^2;
```

$$f := x^2 + y^2 + z^2$$

Soggetta ai seguenti vincoli

```
> v := [x+y+3*z=3,x^2-z^2=0] ;
```

$$v := [x + y + 3\,z = 3, x^2 - z^2 = 0]$$

Soluzione con le primitive Maple

```
> Minimize(f, v);
```

$$[1., [x = 0.666666666644115224, z = 0.666666666644115224, y = 0.333333333423539046]]$$

Uso dei moltiplicatori di Lagrange

```
> v1 := lhs(v[1])-rhs(v[1]) ;
  v2 := lhs(v[2])-rhs(v[2]) ;
```

$$v1 := x + y + 3\,z - 3$$

$$v2 := x^2 - z^2$$

```
> g := f - lambda*v1 - mu*v2 ;
```

$$g := x^2 + y^2 + z^2 - \lambda\,(x + y + 3\,z - 3) - \mu\,(x^2 - z^2)$$

Sistema non lineare da risolvere

```
> F := Gradient(g,[x,y,z,lambda,mu]) ;
```

$$F := (2\,x - \lambda - 2\,\mu\,x)\,\overline{e}_x + (2\,y - \lambda)\,\overline{e}_y + (2\,z - 3\,\lambda + 2\,\mu\,z)\,\overline{e}_z + (-x - y - 3\,z + 3)\,\overline{e}_\lambda + (-x^2 + z^2)\,\overline{e}_\mu$$

```
> RES := solve({F[1],F[2],F[3],F[4],F[5]},{x,y,z,lambda,mu}) ;
```

$$RES := \left\{\mu = \frac{1}{2}, x = \frac{2}{3}, y = \frac{1}{3}, \lambda = \frac{2}{3}, z = \frac{2}{3}\right\},\ \{y = 1, z = 1, \mu = 2, x = -1, \lambda = 2\}$$

Controllo proprietà di minimo

```
> Hf  := Hessian(f,[x,y,z]):
  Hv1 := Hessian(v1,[x,y,z]):
  Hv2 := Hessian(v2,[x,y,z]):
  Hf, Hv1, Hv2 ;
```

$$\begin{bmatrix} 2 & 0 & 0 \\ 0 & 2 & 0 \\ 0 & 0 & 2 \end{bmatrix}, \begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix}, \begin{bmatrix} 2 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & -2 \end{bmatrix}$$

```
> JH := Jacobian([v1,v2],[x,y,z]) ;
  NH := NullSpace(JH) ;
```

$$JH := \begin{bmatrix} 1 & 1 & 3 \\ 2\,x & 0 & -2\,z \end{bmatrix}$$

$$NH := \left\{ \begin{bmatrix} \frac{z}{x} \\ -\frac{z+3\,x}{x} \\ 1 \end{bmatrix} \right\}$$

Controllo minimo/massimo locale primo punto

```
> lambda1 := subs(RES[1],lambda);
  mu1     := subs(RES[1],mu);
```

$$\lambda 1 := \frac{2}{3}$$

$$\mu 1 := \frac{1}{2}$$

```
> Hf1 := simplify(Hf - lambda1. Hv1 - mu1. Hv2) ;
```

$$Hf1 := \begin{bmatrix} 1 & 0 & 0 \\ 0 & 2 & 0 \\ 0 & 0 & 3 \end{bmatrix}$$

Cerco nello spazio dei vincoli:

```
> Z1 := subs(RES[1],op(NH)) ;
```

$$Z1 := \begin{bmatrix} 1 \\ -4 \\ 1 \end{bmatrix}$$

E' positivo per ogni alpha > 0, quindi è un minimo locale

```
> simplify(Transpose(alpha.Z1).Hf1.(alpha.Z1)) ;
```

$$36\,\alpha^2$$

Controllo minimo/massimo locale secondo punto

```
> lambda2 := subs(RES[2],lambda);
  mu2     := subs(RES[2],mu);
```

$$\lambda 2 := 2$$

$$\mu 2 := 2$$

```
> Hf2 := simplify(Hf - lambda2. Hv1 - mu2. Hv2) ;
```

$$Hf2 := \begin{bmatrix} -2 & 0 & 0 \\ 0 & 2 & 0 \\ 0 & 0 & 6 \end{bmatrix}$$

Cerco nello spazio dei vincoli:

```
> Z2 := subs(RES[2],op(NH)) ;
```

$$Z2 := \begin{bmatrix} -1 \\ -2 \\ 1 \end{bmatrix}$$

```
> simplify(Transpose(alpha.Z2).Hf2.(alpha.Z2)) ;
```

$$12\,\alpha^2$$

L'Hessiano è definito positivo: è minimo locale.

Calcolo i valori: il secondo è il minimo assoluto.

```
> subs(RES[1],f);
```

$$1$$

```
> subs(RES[2],f);
```

$$3$$